Example problem

Solve the heat equation $u_{xx} = u_t$ on [0, 1] with boundary conditions $u(0, t) = 0$, $u_x(1, t) = 0$ and initial conditions $u(x, 0) = x(2 - x)$.

Separate variables to find product solutions

Assume a product solution $u(x, t) = X(x)\,T(t)$, plug in to the heat equation, divide by $X\,T$, and derive ordinary differential equations

$$T' = -k^2\,T$$

$$X'' = -k^2\,X$$

The solution to the first equation is $T = A\,e^{-k^2 t}$. The solution to the second is $X(x) = A\cos k\,x + B\sin k\,x$. From the boundary conditions on u, it follows that the boundary conditions on X are $X(0) = 0$, $X'(1) = 0$. Using these BCs we find that

$$X(0) = A = 0$$

so that $A = 0$, and

$$X'(1) = k\,B\cos k = 0$$

so that either $B = 0$ (trivial solution) or $\cos k = 0$. Nontrivial solutions exist only when $k = \frac{\pi}{2}(2\,n - 1)$, $n = 1, 2, 3, \ldots$.

The general product solution is therefore

$$u_n(x, t) = B_n \sin\left(\frac{\pi}{2}(2\,n - 1)\,x\right) e^{-\left(\frac{\pi}{2}(2\,n-1)\right)^2 t}.$$

Superpose product solutions

A linear combination of solutions to the heat equation is also a solution, so we can form a general solution as a linear combination
of product solutions

$$u(x, t) = \sum_{n=1}^{\infty} B_n \sin\left(\frac{\pi}{2}(2\,n - 1)\,x\right) e^{-\left(\frac{\pi}{2}(2\,n-1)\right)^2 t}$$

So far, we've used the heat equation to derive the product solutions and we've used the BCs to eliminate the cosine terms and determine the values of the separation constants k. The remaining undetermined constants are the coefficients B_n.

Use initial conditions to determine coefficients

The initial conditions are $u(x, 0) = x(2 - x) = f(x)$, so that

$$f(x) = \sum_{n=1}^{\infty} B_n \sin\left(\frac{\pi}{2}(2\,n - 1)\,x\right).$$

The functions $\sin\left(\frac{\pi}{2}(2n-1)x\right)$ are an orthogonal set (how do you know this?) so that

$B_n = \frac{<f(x),\phi_n(x)>}{<\phi_n(x),\phi_n(x)>}$.

Do the calculation

Compute the Fourier coefficients of the initial conditions

Define a function for $f(x) = x(2-x)$

In[362]:=
```
f[x_] = x (2 - x)
```

Out[362]= $(2-x)x$

Do the necessary integrations to find the Fourier coefficients. Using the "Assumptions" option to the Integrate function helps simplify the results.

In[364]:=
```
B[n_] = Integrate[f[x] Sin[Pi / 2 (2 n - 1) x], {x, 0, 1}, Assumptions → Element[n, Integers]] /
  Integrate[Sin[ Pi / 2 (2 n - 1) x] ^2, {x, 0, 1}, Assumptions → Element[n, Integers]]
```

Out[364]= $$\frac{32}{(2n-1)^3 \pi^3}$$

Form the *M*th partial sum

In[365]:=
```
fSum[M_, x_] := Sum[B[n] Sin[Pi / 2 (2 n - 1) x], {n, 1, M}]
```

In[368]:=
```
f2[x_] = fSum[2, x]
```

Out[368]= $$\frac{32\sin\left(\frac{\pi x}{2}\right)}{\pi^3} + \frac{32\sin\left(\frac{3\pi x}{2}\right)}{27\pi^3}$$

Plot a partial sum

Plot $f_2(x)$ and compare to the exact $f(x)$

In[369]:=
```
Plot[{f[x], f2[x]}, {x, 0, 1}]
```

Out[369]=



Form a partial sum for the heat solution

In[370]:=
```
uSum[M_, x_, t_] := Sum[B[n] Exp[-(Pi/2)^2 (2 n - 1)^2 t] Sin[Pi/2 (2 n - 1) x], {n, 0, M}]
```

In[371]:=
```
uSum[4, x, t]
```

Out[371]=
$$\frac{64\, e^{-\frac{\pi^2 t}{4}} \sin\left(\frac{\pi x}{2}\right)}{\pi^3} + \frac{32\, e^{-\frac{9\pi^2 t}{4}} \sin\left(\frac{3\pi x}{2}\right)}{27\,\pi^3} + \frac{32\, e^{-\frac{25\pi^2 t}{4}} \sin\left(\frac{5\pi x}{2}\right)}{125\,\pi^3} + \frac{32\, e^{-\frac{49\pi^2 t}{4}} \sin\left(\frac{7\pi x}{2}\right)}{343\,\pi^3}$$

Plug into the heat equation to check. The result should be zero.

In[372]:=
```
D[uSum[4, x, t], {x, 2}] - D[uSum[4, x, t], t]
```

Out[372]= 0

■ Do a surface plot of a partial sum of the solution

In[373]:=
```
Plot3D[uSum[4, x, t], {x, 0, 1}, {t, 0, 1}]
```

Out[373]=